Pensieve header: The double and meta-double of the 2D pencil; continues pensieve://2017-04/, continued pensieve://2017-06/.

Issues:

1. S does not invert R. (Perhaps because H must be interpreted as $e^{\gamma h}$).
2. dm is not meta-associative.
3. R doesn't satisfy YB.

```
SetDirectory["C:\\drorbn\\AcademicPensieve\\2017-05"];
```

## The 2D Lie BiAlgebra Pencil

I hope to stick to $G = e^{\eta g}$ and to $H = e^{\gamma h}$, where $[g, e] = \gamma\, e$ and $[h, f] = -\eta\, f$.
Also, $q\Delta_{12}(g, G, e, h, H, f) = (g_1 + g_2,\ G_1\, G_2,\ e_1 + G_1\, e_2,\ h_1 + h_2,\ H_1\, H_2,\ f_1\, H_2 + f_2)$.
Also, $(g, e)$ and $(h, f)$ are dual bases.

```
AlgebraAtom = g | G[_] | e | h | H[_] | f;
$PBWRule = {G[_] → 1, g → 2, e → 3, H[_] → 4, h → 5, f → 6};
```

```
B[g, e] = γ e; B[e, G[n_]] = (e^(-n γ η) - 1) U[G[n], e]; B[g, G[_]] = 0;
B[h, f] = -η f; B[f, H[n_]] = (e^(n γ η) - 1) U[H[n], f]; B[h, H[_]] = 0;
```

## UEA with provisional modification

This section is based on penseive://Projects/UEA/.

```
B[0, _] = 0; B[_, 0] = 0;
B[c_ * x : AlgebraAtom, y_] := Expand[c B[x, y]];
B[y_, c_ * x : AlgebraAtom] := Expand[c B[y, x]];
B[x_Plus, y_] := B[#, y] & /@ x;
B[x_, y_Plus] := B[x, #] & /@ y;
B[x_, x_] = 0;
B[y_, x_] := Expand[-B[x, y]];
```

```
x_ ≼ y_ := OrderedQ[{x, y} /. $PBWRule]; x_ ≺ y_ := ! OrderedQ[{y, x} /. $PBWRule];
UU_i_[] := U_i[]; UU_i_[1] := U_i[];
UU_i_[x_[n_]^p_] := U_i[x[n p]];
UU_i_[x_^p_] := UU_i @@ Table[x, {p}];
UU_i_[ℰ_] := ℰ /. {
    U[xs__] :⇢ U_i[xs],
    x : AlgebraAtom :⇢ U_i[x]
   };
UU_i_[x_, xs__] := UU_t1[x] UU_t2[xs] // Expand // m_{t1,t2→i};
USimp[sd_SeriesData] := MapAt[USimp, sd, {3, All}];
USimp[ℰ_] := Collect[ℰ, Times[U_[__] ..], Expand];
USimp[ℰ_] := Expand[ℰ];
```

```
m_s__[0] = 0;
m_s__[x_Plus] := m_s /@ x;
m_s__[sd_SeriesData] := MapAt[m_s, sd, {3, All}];
m_i_→j_[ℰ_] := ℰ /. U_i → U_j;
```

```
m_i_,j_→k_[c_. U_i_[x___] U_j_[]] := c U_k[x];
m_i_,j_→k_[c_. U_i_[] U_j_[y___]] := c U_k[y];
m_i_,j_→k_[c_. U_i_[xx___, x_[n1_]] U_j_[x_[n2_], yy___]] :=
  USimp[c If[n1 + n2 == 0, U_i[xx] U_j[yy], U_i[xx, x[n1 + n2]] U_j[yy]] // m_{i,j→k}];
m_i_,j_→k_[c_. U_i_[xx___, x_] U_j_[y_, yy___]] := If[x ≼ y,
   c U_k[xx, x, y, yy],
   ((U_i[xx] (U_j[y, x] + UU_j[B[x, y]]) // Expand // m_{i,j→i}) U_j[yy] // Expand // m_{i,j→k})
     c // USimp
  ];
```

```
Supp[ℰ_] := Union@Cases[{ℰ}, U_i_[___] :⇢ i, ∞];
```

```
Unprotect[NonCommutativeMultiply];
NonCommutativeMultiply[x_] := x;
x_ ** y_ := Module[
   {Sx = Supp[x], Sy = Supp[y], is, σ, z},
   If[MatchQ[Sx ⋃ Sy, {_Integer ...}] && Min[Sx ⋃ Sy] < 0,
    is = Abs[Sx] ⋂ Abs[Sy];
    z = x; Do[z = m_{-i→-σ@i}[m_{i→σ@i}[z]], {i, is}];
    z = USimp[y z]; Do[z = dm_{σ@i,i→i}[z], {i, is}];
    z,
    (* else *) is = Sx ⋂ Sy;
    z = x; Do[z = m_{i→σ@i}[z], {i, is}];
    z = USimp[y z]; Do[z = m_{σ@i,i→i}[z], {i, is}];
    z
   ]
  ];
UB[x_, y_] := USimp[x ** y - y ** x];
```

```
𝕆[specs_, sd_SeriesData] := MapAt[𝕆[specs, #] &, sd, {3, All}];
𝕆[specs_, poly_] := Module[{rules, vars, elems},
  rules = Union @@ Cases[{specs}, U_[u___] :> Cases[{u}, r_Rule], ∞];
  vars = First /@ rules; elems = Last /@ rules;
  USimp@Total[CoefficientRules[poly, vars] /. (ps_ → c_) :> c (
       specs /. MapThread[((#1 → _) :> #3^#2) &, {vars, ps, elems}] /. U_i_ :> UU_i
      )]
 ]
```

## The 2D Lie BiAlgebra Pencil, Testing

```
𝕆[U₁[x → g], Normal@Series[e^(η x), {η, 0, 5}]]
```

$$U_1[\,] + \eta\, U_1[g] + \frac{1}{2}\eta^2\, U_1[g, g] + \frac{1}{6}\eta^3\, U_1[g, g, g] + \frac{1}{24}\eta^4\, U_1[g, g, g, g] + \frac{1}{120}\eta^5\, U_1[g, g, g, g, g]$$

```
With[{G = 𝕆[U₁[x → g], Series[e^(η x), {η, 0, 5}]]}, UB[U₁[e], G] - (e^(-γ η) - 1) G ** U₁[e]]
```

$$O[\eta]^6$$

```
B[e, G[3]]
```

$$\left(-1 + e^{-3\gamma\eta}\right) U[G[3], e]$$

```
With[{H = 𝕆[U₁[x → h], Series[e^(γ x), {γ, 0, 5}]]}, UB[U₁[f], H] - (e^(γ η) - 1) H ** U₁[f]]
```

$$O[\gamma]^6$$

```
x = U₁[g, G[2], e, e, e] U₂[g, g, e] U₃[g, g, G[-3], e];
(x // m_{1,2→1} // m_{1,3→1}) - (x // m_{2,3→2} // m_{1,2→1})
```

$$0$$

```
x = U₁[h, H[2], f, f, f] U₂[h, h, f] U₃[h, h, H[-3], f];
(x // m_{1,2→1} // m_{1,3→1}) - (x // m_{2,3→2} // m_{1,2→1})
```

$$0$$

## The Co-Product and Co-Associativity

```
qΔ_→_,_[ℰ_] := USimp@Module[{tj, tk}, ℰ /. {
      U_[] → U_[] U_[],
      U_[g, xs___] :→
       (USimp[(U_[g] U_[] + U_[] U_[g]) qΔ_→tj,tk[U_[xs]]] // m_,tj→_ // m_,tk→_),
      U_[G[n_], xs___] :→ (USimp[U_[G[n]] U_[G[n]] qΔ_→tj,tk[U_[xs]]] //
          m_,tj→_ // m_,tk→_),
      U_[e, xs___] :→ (USimp[(U_[e] U_[G[1]] + U_[] U_[e]) qΔ_→tj,tk[U_[xs]]] //
          m_,tj→_ // m_,tk→_),
      U_[h, xs___] :→ (USimp[(U_[h] U_[] + U_[] U_[h]) qΔ_→tj,tk[U_[xs]]] //
          m_,tj→_ // m_,tk→_),
      U_[H[n_], xs___] :→ (USimp[U_[H[n]] U_[H[n]] qΔ_→tj,tk[U_[xs]]] //
          m_,tj→_ // m_,tk→_),
      U_[f, xs___] :→ (USimp[(U_[f] U_[] + U_[H[1]] U_[f]) qΔ_→tj,tk[U_[xs]]] //
          m_,tj→_ // m_,tk→_)
     }];
```

```
qΔ_→_,_,_[ℰ_] := ℰ // qΔ_→_,_ // qΔ_→_,_
```

```
U1[e] // qΔ1→1,2
```

$$U_1[]\,U_2[e] + U_1[e]\,U_2[G[1]]$$

```
{lhs = U1[e] // qΔ1→1,2 // qΔ2→2,3, rhs = U1[e] // qΔ1→1,3 // qΔ1→1,2, lhs == rhs}
```

$$\{U_1[]\,U_2[]\,U_3[e] + U_1[]\,U_2[e]\,U_3[G[1]] + U_1[e]\,U_2[G[1]]\,U_3[G[1]],\ U_1[]\,U_2[]\,U_3[e] + U_1[]\,U_2[e]\,U_3[G[1]] + U_1[e]\,U_2[G[1]]\,U_3[G[1]],\ \text{True}\}$$

```
U1[f] // qΔ1→1,2
```

$$U_1[f]\,U_2[] + U_1[H[1]]\,U_2[f]$$

```
{lhs = U1[f] // qΔ1→1,2 // qΔ2→2,3, rhs = U1[f] // qΔ1→1,3 // qΔ1→1,2, lhs == rhs}
```

$$\{U_1[f]\,U_2[]\,U_3[] + U_1[H[1]]\,U_2[f]\,U_3[] + U_1[H[1]]\,U_2[H[1]]\,U_3[f],\ U_1[f]\,U_2[]\,U_3[] + U_1[H[1]]\,U_2[f]\,U_3[] + U_1[H[1]]\,U_2[H[1]]\,U_3[f],\ \text{True}\}$$

```
x = U1[g, G[2], e, e, e] U2[g, g, G[-3], e];
(x // m1,2→1 // qΔ1→1,2) - (x // qΔ2→3,4 // qΔ1→1,2 // m1,3→1 // m2,4→2)
```

```
0
```

```
x = U1[h, H[2], f, f, f] U2[h, h, H[-3], f];
(x // m1,2→1 // qΔ1→1,2) - (x // qΔ2→3,4 // qΔ1→1,2 // m1,3→1 // m2,4→2)
```

```
0
```

## The Antipode

Why o why this annoyance of left-vs-right?

```
S[g] = -g; S[G[n_]] := G[-n]; S[e] = -𝕖^(γ η) U[G[-1], e];
S[h] = -h; S[H[n_]] := H[-n]; S[f] = -U[H[-1], f];
S_i_[ℰ_] := Module[{ti}, USimp[
     ℰ /. U_i[x_, xs___] :> m_{ti,i→i}[Expand[UU_i[S[x]] S_ti[U_ti[xs]]]]
    ]];
```

```
{lhs = S₁[U₁[e]], rhs = -U₁[e] ** U₁[G[-1]], lhs == rhs}
```

$$\left\{-\mathbb{e}^{\gamma\eta}\,U_1[G[-1],e],\ -\mathbb{e}^{\gamma\eta}\,U_1[G[-1],e],\ \text{True}\right\}$$

```
U₁[e] // S₁ // S₁
```

$$\mathbb{e}^{\gamma\eta}\,U_1[e]$$

```
U₁[f] // S₁ // S₁
```

$$\mathbb{e}^{\gamma\eta}\,U_1[f]$$

```
S₁[U₁[g, G[3], e, e]]
```

$$2\,\mathbb{e}^{9\gamma\eta}\,\gamma\,U_1[G[-5],e,e]-\mathbb{e}^{9\gamma\eta}\,U_1[G[-5],g,e,e]$$

```
U₁[g, G[3], e, e] // qΔ₁→₁,₂
```

$$\begin{aligned}&U_1[G[3],g,e,e]\,U_2[G[5]]+U_1[G[3],g,e]\,U_2[G[4],e]+\\&\mathbb{e}^{-\gamma\eta}\,U_1[G[3],g,e]\,U_2[G[4],e]+U_1[G[3],e,e]\,U_2[G[5],g]+U_1[G[3],g]\,U_2[G[3],e,e]+\\&U_1[G[3],e]\,U_2[G[4],g,e]+\mathbb{e}^{-\gamma\eta}\,U_1[G[3],e]\,U_2[G[4],g,e]+U_1[G[3]]\,U_2[G[3],g,e,e]\end{aligned}$$

```
U₁[g, G[3], e, e] // qΔ₁→₁,₂ // S₂
```

$$\begin{aligned}&U_1[G[3],g,e,e]\,U_2[G[-5]]-\mathbb{e}^{4\gamma\eta}\,\gamma\,U_1[G[3],e]\,U_2[G[-5],e]-\mathbb{e}^{5\gamma\eta}\,\gamma\,U_1[G[3],e]\,U_2[G[-5],e]-\\&\mathbb{e}^{4\gamma\eta}\,U_1[G[3],g,e]\,U_2[G[-5],e]-\mathbb{e}^{5\gamma\eta}\,U_1[G[3],g,e]\,U_2[G[-5],e]-\\&U_1[G[3],e,e]\,U_2[G[-5],g]+2\,\mathbb{e}^{9\gamma\eta}\,\gamma\,U_1[G[3]]\,U_2[G[-5],e,e]+\\&\mathbb{e}^{9\gamma\eta}\,U_1[G[3],g]\,U_2[G[-5],e,e]+\mathbb{e}^{4\gamma\eta}\,U_1[G[3],e]\,U_2[G[-5],g,e]+\\&\mathbb{e}^{5\gamma\eta}\,U_1[G[3],e]\,U_2[G[-5],g,e]-\mathbb{e}^{9\gamma\eta}\,U_1[G[3]]\,U_2[G[-5],g,e,e]\end{aligned}$$

```
test = U₁[g, G[3], e, e];
{test // qΔ₁→₁,₂ // S₂ // m₁,₂→₁, test // qΔ₁→₁,₂ // S₂ // m₂,₁→₁,
 test // qΔ₁→₁,₂ // S₁ // m₁,₂→₁, test // qΔ₁→₁,₂ // S₁ // m₂,₁→₁}
```

$$\left\{0,\ 0,\ 0,\ 2\,\mathbb{e}^{6\gamma\eta}\,\gamma\,U_1[e,e]-2\,\mathbb{e}^{7\gamma\eta}\,\gamma\,U_1[e,e]-2\,\mathbb{e}^{8\gamma\eta}\,\gamma\,U_1[e,e]+2\,\mathbb{e}^{9\gamma\eta}\,\gamma\,U_1[e,e]\right\}$$

```
test = U₁[h, H[3], f, f];
{test // qΔ₁→₁,₂ // S₂ // m₁,₂→₁, test // qΔ₁→₁,₂ // S₂ // m₂,₁→₁,
 test // qΔ₁→₁,₂ // S₁ // m₁,₂→₁, test // qΔ₁→₁,₂ // S₁ // m₂,₁→₁}
```

$$\begin{aligned}\{&0,\ 0,\ 0,\ -2\,\mathbb{e}^{-10\gamma\eta}\,\eta\,U_1[H[-2],f,f]+\\&2\,\mathbb{e}^{-9\gamma\eta}\,\eta\,U_1[H[-2],f,f]+2\,\mathbb{e}^{-8\gamma\eta}\,\eta\,U_1[H[-2],f,f]-2\,\mathbb{e}^{-7\gamma\eta}\,\eta\,U_1[H[-2],f,f]\}\end{aligned}$$

```
x = U₁[h, H[2], f, f, f] U₂[h, h, H[-3], f];
(x // m₁,₂→₁ // S₁) - (x // S₁ // S₂ // m₂,₁→₁)
```

$$0$$

```
x = U1[g, G[2], e, e, e] U2[g, g, G[-3], e];
(x // m1,2→1 // S1) - (x // S1 // S2 // m2,1→1)
```
```
0
```

```
x = U1[];
(x // qΔ1→1,2 // S1 // m1,2→1)
```
```
U1[]
```

```
x = U1[];
(x // qΔ1→1,2 // S2 // m1,2→1)
```
```
U1[]
```

```
x = U1[g, G[2], e, e, e];
(x // qΔ1→1,2 // S1 // m1,2→1)
```
```
0
```

```
x = U1[g, G[2], e, e, e];
(x // qΔ1→1,2 // S2 // m1,2→1)
```
```
0
```

```
x = U1[h, H[2], f, f, f];
(x // qΔ1→1,2 // S1 // m1,2→1)
```
```
0
```

```
x = U1[h, H[2], f, f, f];
(x // qΔ1→1,2 // S2 // m1,2→1)
```
```
0
```

## The Pairing at Lie-Level and Compatibilities

```
P[U[], U[]] = 1;
P[U[], U[H[_]]] = P[U[G[_]], U[]] = 1;
P[U[], U[__]] = P[U[__], U[]] = 0;
{ P[U[g], U[h]] = 1      P[U[g], U[H[n_]]] = n γ      P[U[g], U[f]] = 0
  P[U[G[n_]], U[h]] = n η  P[U[G[n_]], U[H[m_]]] = e^(n m η γ)  P[U[G[_]], U[f]] = 0
  P[U[e], U[h]] = 0      P[U[e], U[H[_]]] = 0      P[U[e], U[f]] = 1 };
```

```
Pi_,j_[ℰ_] := USimp[ℰ /. Ui[xs___] Uj[ys___] → P[U[xs], U[ys]]];
```

```
t = Ui[g] Uj[e] Uk[f];
{mi,j→i[t] - mj,i→i[t], qΔk→k,1[t] - qΔk→1,k[t]}
```
```
{γ Ui[e] Uk[f], Ui[g] Uj[e] Uk[f] Ul[] - Ui[g] Uj[e] Uk[] Ul[f] +
  Ui[g] Uj[e] Uk[H[1]] Ul[f] - Ui[g] Uj[e] Uk[f] Ul[H[1]]}
```

```
t = U_i[g] U_j[e] U_k[f];
{(m_{i,j→i}[t] - m_{j,i→i}[t]) // P_{i,k}, (qΔ_{k→k,l}[t] - qΔ_{k→l,k}[t]) // P_{i,k} // P_{j,l}}
```

$$\{\gamma, \gamma\}$$

```
Table[t = U_i[xi] U_j[xj] U_k[yk];
 {(m_{i,j→i}[t] - m_{j,i→i}[t]) // P_{i,k}, (qΔ_{k→k,l}[t] - qΔ_{k→l,k}[t]) // P_{i,k} // P_{j,l}},
 {xi, {g, e}}, {xj, {g, e}}, {yk, {h, f}}]
```

$$\{\{\{\{0, 0\}, \{0, 0\}\}, \{\{0, 0\}, \{\gamma, \gamma\}\}\}, \{\{\{0, 0\}, \{-\gamma, -\gamma\}\}, \{\{0, 0\}, \{0, 0\}\}\}\}$$

```
Table[t = U_i[xi] U_k[yk] U_l[yl];
 {(qΔ_{i→i,j}[t] - qΔ_{i→j,i}[t]) // P_{i,k} // P_{j,l}, (m_{k,l→k}[t] - m_{l,k→k}[t]) // P_{i,k}},
 {xi, {g, e}}, {yk, {h, f}}, {yl, {h, f}}]
```

$$\{\{\{\{0, 0\}, \{0, 0\}\}, \{\{0, 0\}, \{0, 0\}\}\}, \{\{\{0, 0\}, \{-\eta, -\eta\}\}, \{\{\eta, \eta\}, \{0, 0\}\}\}\}$$

## General Pairings

The pairing sequence: ⟨one,one⟩ (above), ⟨many,one⟩, ⟨many,many⟩.

```
P[U[x_, xs__], U[y_]] := P[U[x, xs], U[y]] =
   Module[{i, j, k, l}, USimp[U_i[x] U_j[xs] qΔ_{k→k,l}[U_k[y]]] // P_{i,k} // P_{j,l}];
P[U[xs__], U[y_, ys__]] := P[U[xs], U[y, ys]] =
   Module[{i, j, k, l}, USimp[qΔ_{i→i,j}[U_i[xs]] U_k[y] U_l[ys]] // P_{i,k} // P_{j,l}];
```

```
{P[U[g, e], U[h]], P[U[g, e], U[f]], P[U[e, e], U[f]]}
```

$$\{0, \gamma, 0\}$$

```
P[U[e], U[f, f]]
```

$$0$$

```
P[U[e, e], U[f, f]]
```

$$1 + e^{\gamma\eta}$$

```
lhs = Factor@Table[P[U @@ Table[e, {n}], U @@ Table[f, {n}]], {n, 7}]
```

$$\left\{1, 1 + e^{\gamma\eta}, \left(1 + e^{\gamma\eta}\right)\left(1 + e^{\gamma\eta} + e^{2\gamma\eta}\right), \left(1 + e^{\gamma\eta}\right)^2\left(1 + e^{2\gamma\eta}\right)\left(1 + e^{\gamma\eta} + e^{2\gamma\eta}\right), \left(1 + e^{\gamma\eta}\right)^2\left(1 + e^{2\gamma\eta}\right)\left(1 + e^{\gamma\eta} + e^{2\gamma\eta} + e^{3\gamma\eta} + e^{4\gamma\eta}\right), \left(1 + e^{\gamma\eta}\right)^3\left(1 + e^{2\gamma\eta}\right)\left(1 - e^{\gamma\eta} + e^{2\gamma\eta}\right)\left(1 + e^{\gamma\eta} + e^{2\gamma\eta}\right)^2\left(1 + e^{\gamma\eta} + e^{2\gamma\eta} + e^{3\gamma\eta} + e^{4\gamma\eta}\right), \left(1 + e^{\gamma\eta}\right)^3\left(1 + e^{2\gamma\eta}\right)\left(1 - e^{\gamma\eta} + e^{2\gamma\eta}\right)\left(1 + e^{\gamma\eta} + e^{2\gamma\eta}\right)^2 \left(1 + e^{\gamma\eta} + e^{2\gamma\eta} + e^{3\gamma\eta} + e^{4\gamma\eta}\right)\left(1 + e^{\gamma\eta} + e^{2\gamma\eta} + e^{3\gamma\eta} + e^{4\gamma\eta} + e^{5\gamma\eta} + e^{6\gamma\eta}\right)\right\}$$

```
rhs = Simplify@FunctionExpand@Table[QFactorial[n, e^{γ η}], {n, 7}]
```

$$\left\{1, 1 + e^{\gamma\eta}, \left(1 + e^{\gamma\eta}\right)\left(1 + e^{\gamma\eta} + e^{2\gamma\eta}\right), \left(1 + e^{\gamma\eta}\right)^2\left(1 + e^{2\gamma\eta}\right)\left(1 + e^{\gamma\eta} + e^{2\gamma\eta}\right), \left(1 + e^{\gamma\eta}\right)^2\left(1 + e^{2\gamma\eta}\right)\left(1 + e^{\gamma\eta} + e^{2\gamma\eta} + e^{3\gamma\eta} + e^{4\gamma\eta}\right), \left(1 + e^{\gamma\eta}\right)^3\left(1 + e^{2\gamma\eta}\right)\left(1 - e^{\gamma\eta} + e^{2\gamma\eta}\right)\left(1 + e^{\gamma\eta} + e^{2\gamma\eta}\right)^2\left(1 + e^{\gamma\eta} + e^{2\gamma\eta} + e^{3\gamma\eta} + e^{4\gamma\eta}\right), \left(1 + e^{\gamma\eta}\right)^3\left(1 + e^{2\gamma\eta}\right)\left(1 - e^{\gamma\eta} + e^{2\gamma\eta}\right)\left(1 + e^{\gamma\eta} + e^{2\gamma\eta}\right)^2 \left(1 + e^{\gamma\eta} + e^{2\gamma\eta} + e^{3\gamma\eta} + e^{4\gamma\eta}\right)\left(1 + e^{\gamma\eta} + e^{2\gamma\eta} + e^{3\gamma\eta} + e^{4\gamma\eta} + e^{5\gamma\eta} + e^{6\gamma\eta}\right)\right\}$$

```
lhs == rhs
```
```
True
```

```
P[U[g, g, g, g, g], U[h, h, h, h, h]]
```
```
120
```

```
P[U[g, g, g, g, g, e, e, e, e], U[h, h, h, h, h, f, f, f, f]] // Factor
```
$$120\left(1+e^{\gamma\eta}\right)^2\left(1+e^{2\gamma\eta}\right)\left(1+e^{\gamma\eta}+e^{2\gamma\eta}\right)$$

```
x = U1[g, G[-3], e, e] U2[g, g, G[1], e] U3[h, H[2], f, f, f];
(x // m1,2→1 // P1,3) - (x // qΔ3→3,4 // P1,3 // P2,4)
```
```
0
```

```
x = U1[h, H[-3], f, f] U2[h, h, H[1], f] U3[g, G[2], e, e, e];
(x // m1,2→1 // P3,1) - (x // qΔ3→3,4 // P3,1 // P4,2)
```
```
0
```

```
x = U1[g, g, G[-3], e, e, e] U2[h, h, H[2], f, f, f];
(x // S1 // P1,2) - (x // S2 // P1,2)
```
```
0
```

## The Double

```
dmi_,j_→k_[ℰ_] := Module[{t1, t2, t3, h1, h2, h3},
  ℰ // qΔi→h1,h2,h3 // Sh1 // qΔ-j→t1,t2,t3 // Ph1,t1 // Ph3,t3 // mh2,j→k // m-i,t2→-k]
```

```
U-1[] U1[g] U-2[h] U2[] // dm1,2→1
```
```
U-1[h] U1[g]
```

```
U-1[] U1[g] U-2[f] U2[] // dm1,2→1
```
$$-\gamma\,U_{-1}[f]\,U_1[\,]+U_{-1}[f]\,U_1[g]$$

```
U-1[] U1[G[1]] U-2[f] U2[] // dm1,2→1
```
$$e^{-\gamma\eta}\,U_{-1}[f]\,U_1[G[1]]$$

```
U-1[] U1[e] U-2[h] U2[] // dm1,2→1
```
$$\eta\,U_{-1}[\,]\,U_1[e]+U_{-1}[h]\,U_1[e]$$

```
U-1[] U1[e] U-2[H[1]] U2[] // dm1,2→1
```
$$e^{\gamma\eta}\,U_{-1}[H[1]]\,U_1[e]$$

```
3□ U-1[] U1[e] U-2[f] U2[] // dm1,2→1
```
$$3_\square\,U_{-1}[H[1]]\,U_1[\,]+3_\square\,U_{-1}[f]\,U_1[e]-3_\square\,U_{-1}[\,]\,U_1[G[1]]$$

```
x = U_-1[] U_1[e] U_-2[f] U_2[] U_-3[h] U_3[];
(x // dm_{1,2→1} // dm_{1,3→1}) - (x // dm_{2,3→2} // dm_{1,2→1})
```

$$-\eta\, U_{-1}[H[1]]\, U_1[\,] + \eta\, U_{-1}[\,]\, U_1[G[1]]$$

```
x = U_-1[h, h, f] U_1[G[2], e, e] U_-2[H[-1], f, f] U_2[g] U_-3[h, f] U_3[g, g, e];
(x // dm_{1,2→1} // dm_{1,3→1}) - (x // dm_{2,3→2} // dm_{1,2→1})
```

$$\begin{aligned}
&2\,e^{-5\gamma\eta}\gamma^3\eta\, U_{-1}[h,h,f,f,f]\, U_1[G[2],e,e] - 4\,e^{-4\gamma\eta}\gamma^3\eta\, U_{-1}[h,h,f,f,f]\, U_1[G[2],e,e] +\\
&2\,e^{-3\gamma\eta}\gamma^3\eta\, U_{-1}[h,h,f,f,f]\, U_1[G[2],e,e] + 8\,e^{-2\gamma\eta}\gamma^3\eta\, U_{-1}[h,h,f,f,f]\, U_1[G[2],e,e] -\\
&2\,e^{-4\gamma\eta}\gamma^3\, U_{-1}[h,h,h,f,f,f]\, U_1[G[2],e,e] + 2\,e^{-2\gamma\eta}\gamma^3\, U_{-1}[h,h,h,f,f,f]\, U_1[G[2],e,e] -\\
&2\,e^{-9\gamma\eta}\gamma^3\eta\, U_{-1}[H[-1],h,h,f,f,f]\, U_1[G[3],e,e] +\\
&4\,e^{-8\gamma\eta}\gamma^3\eta\, U_{-1}[H[-1],h,h,f,f,f]\, U_1[G[3],e,e] -\\
&2\,e^{-7\gamma\eta}\gamma^3\eta\, U_{-1}[H[-1],h,h,f,f,f]\, U_1[G[3],e,e] -\\
&8\,e^{-6\gamma\eta}\gamma^3\eta\, U_{-1}[H[-1],h,h,f,f,f]\, U_1[G[3],e,e] +\\
&2\,e^{-8\gamma\eta}\gamma^3\, U_{-1}[H[-1],h,h,h,f,f,f]\, U_1[G[3],e,e] -\\
&2\,e^{-6\gamma\eta}\gamma^3\, U_{-1}[H[-1],h,h,h,f,f,f]\, U_1[G[3],e,e] -\\
&5\,e^{-5\gamma\eta}\gamma^2\eta\, U_{-1}[h,h,f,f,f]\, U_1[G[2],g,e,e] +\\
&10\,e^{-4\gamma\eta}\gamma^2\eta\, U_{-1}[h,h,f,f,f]\, U_1[G[2],g,e,e] -\\
&5\,e^{-3\gamma\eta}\gamma^2\eta\, U_{-1}[h,h,f,f,f]\, U_1[G[2],g,e,e] -\\
&20\,e^{-2\gamma\eta}\gamma^2\eta\, U_{-1}[h,h,f,f,f]\, U_1[G[2],g,e,e] +\\
&5\,e^{-4\gamma\eta}\gamma^2\, U_{-1}[h,h,h,f,f,f]\, U_1[G[2],g,e,e] -\\
&5\,e^{-2\gamma\eta}\gamma^2\, U_{-1}[h,h,h,f,f,f]\, U_1[G[2],g,e,e] +\\
&\gamma\eta\, U_{-1}[H[1],h,h,f,f]\, U_1[G[2],g,g,e] + 4\,e^{\gamma\eta}\gamma\eta\, U_{-1}[H[1],h,h,f,f]\, U_1[G[2],g,g,e] +\\
&3\,e^{2\gamma\eta}\gamma\eta\, U_{-1}[H[1],h,h,f,f]\, U_1[G[2],g,g,e] -\\
&\gamma\, U_{-1}[H[1],h,h,h,f,f]\, U_1[G[2],g,g,e] - e^{-\gamma\eta}\gamma\, U_{-1}[H[1],h,h,h,f,f]\, U_1[G[2],g,g,e] +\\
&e^{\gamma\eta}\gamma\, U_{-1}[H[1],h,h,h,f,f]\, U_1[G[2],g,g,e] +\\
&e^{2\gamma\eta}\gamma\, U_{-1}[H[1],h,h,h,f,f]\, U_1[G[2],g,g,e] +\\
&5\,e^{-9\gamma\eta}\gamma^2\eta\, U_{-1}[H[-1],h,h,f,f,f]\, U_1[G[3],g,e,e] -\\
&10\,e^{-8\gamma\eta}\gamma^2\eta\, U_{-1}[H[-1],h,h,f,f,f]\, U_1[G[3],g,e,e] +\\
&5\,e^{-7\gamma\eta}\gamma^2\eta\, U_{-1}[H[-1],h,h,f,f,f]\, U_1[G[3],g,e,e] +\\
&20\,e^{-6\gamma\eta}\gamma^2\eta\, U_{-1}[H[-1],h,h,f,f,f]\, U_1[G[3],g,e,e] -\\
&5\,e^{-8\gamma\eta}\gamma^2\, U_{-1}[H[-1],h,h,h,f,f,f]\, U_1[G[3],g,e,e] +\\
&5\,e^{-6\gamma\eta}\gamma^2\, U_{-1}[H[-1],h,h,h,f,f,f]\, U_1[G[3],g,e,e] - 3\gamma\eta\, U_{-1}[h,h,f,f]\, U_1[G[3],g,g,e] -\\
&e^{-3\gamma\eta}\gamma\eta\, U_{-1}[h,h,f,f]\, U_1[G[3],g,g,e] - 5\,e^{-2\gamma\eta}\gamma\eta\, U_{-1}[h,h,f,f]\, U_1[G[3],g,g,e] -\\
&7\,e^{-\gamma\eta}\gamma\eta\, U_{-1}[h,h,f,f]\, U_1[G[3],g,g,e] - \gamma\, U_{-1}[h,h,h,f,f]\, U_1[G[3],g,g,e] +\\
&e^{-4\gamma\eta}\gamma\, U_{-1}[h,h,h,f,f]\, U_1[G[3],g,g,e] + 2\,e^{-3\gamma\eta}\gamma\, U_{-1}[h,h,h,f,f]\, U_1[G[3],g,g,e] -\\
&2\,e^{-\gamma\eta}\gamma\, U_{-1}[h,h,h,f,f]\, U_1[G[3],g,g,e] + e^{-5\gamma\eta}\gamma\eta\, U_{-1}[H[-1],h,h,f,f]\, U_1[G[4],g,g,e] +\\
&4\,e^{-4\gamma\eta}\gamma\eta\, U_{-1}[H[-1],h,h,f,f]\, U_1[G[4],g,g,e] +\\
&3\,e^{-3\gamma\eta}\gamma\eta\, U_{-1}[H[-1],h,h,f,f]\, U_1[G[4],g,g,e] -\\
&e^{-6\gamma\eta}\gamma\, U_{-1}[H[-1],h,h,h,f,f]\, U_1[G[4],g,g,e] -\\
&e^{-5\gamma\eta}\gamma\, U_{-1}[H[-1],h,h,h,f,f]\, U_1[G[4],g,g,e] +\\
&e^{-4\gamma\eta}\gamma\, U_{-1}[H[-1],h,h,h,f,f]\, U_1[G[4],g,g,e] +\\
&e^{-3\gamma\eta}\gamma\, U_{-1}[H[-1],h,h,h,f,f]\, U_1[G[4],g,g,e] +\\
&4\,e^{-5\gamma\eta}\gamma\eta\, U_{-1}[h,h,f,f,f]\, U_1[G[2],g,g,e,e] -\\
&8\,e^{-4\gamma\eta}\gamma\eta\, U_{-1}[h,h,f,f,f]\, U_1[G[2],g,g,e,e] +\\
&4\,e^{-3\gamma\eta}\gamma\eta\, U_{-1}[h,h,f,f,f]\, U_1[G[2],g,g,e,e] +\\
&16\,e^{-2\gamma\eta}\gamma\eta\, U_{-1}[h,h,f,f,f]\, U_1[G[2],g,g,e,e] -\\
&4\,e^{-4\gamma\eta}\gamma\, U_{-1}[h,h,h,f,f,f]\, U_1[G[2],g,g,e,e] +\\
&4\,e^{-2\gamma\eta}\gamma\, U_{-1}[h,h,h,f,f,f]\, U_1[G[2],g,g,e,e] -\\
&\eta\, U_{-1}[H[1],h,h,f,f]\, U_1[G[2],g,g,g,e] - 4\,e^{\gamma\eta}\eta\, U_{-1}[H[1],h,h,f,f]\, U_1[G[2],g,g,g,e] -\\
&3\,e^{2\gamma\eta}\eta\, U_{-1}[H[1],h,h,f,f]\, U_1[G[2],g,g,g,e] +
\end{aligned}$$

$$
\begin{aligned}
&U_{-1}[H[1],h,h,h,f,f]\,U_1[G[2],g,g,g,e]+\\
&e^{-\gamma\eta}\,U_{-1}[H[1],h,h,h,f,f]\,U_1[G[2],g,g,g,e]-\\
&e^{\gamma\eta}\,U_{-1}[H[1],h,h,h,f,f]\,U_1[G[2],g,g,g,e]-\\
&e^{2\gamma\eta}\,U_{-1}[H[1],h,h,h,f,f]\,U_1[G[2],g,g,g,e]-\\
&4\,e^{-9\gamma\eta}\,\gamma\,\eta\,U_{-1}[H[-1],h,h,f,f,f]\,U_1[G[3],g,g,e,e]+\\
&8\,e^{-8\gamma\eta}\,\gamma\,\eta\,U_{-1}[H[-1],h,h,f,f,f]\,U_1[G[3],g,g,e,e]-\\
&4\,e^{-7\gamma\eta}\,\gamma\,\eta\,U_{-1}[H[-1],h,h,f,f,f]\,U_1[G[3],g,g,e,e]-\\
&16\,e^{-6\gamma\eta}\,\gamma\,\eta\,U_{-1}[H[-1],h,h,f,f,f]\,U_1[G[3],g,g,e,e]+\\
&4\,e^{-8\gamma\eta}\,\gamma\,U_{-1}[H[-1],h,h,h,f,f,f]\,U_1[G[3],g,g,e,e]-\\
&4\,e^{-6\gamma\eta}\,\gamma\,U_{-1}[H[-1],h,h,h,f,f,f]\,U_1[G[3],g,g,e,e]+\\
&3\,\eta\,U_{-1}[h,h,f,f]\,U_1[G[3],g,g,g,e]+e^{-3\gamma\eta}\,\eta\,U_{-1}[h,h,f,f]\,U_1[G[3],g,g,g,e]+\\
&5\,e^{-2\gamma\eta}\,\eta\,U_{-1}[h,h,f,f]\,U_1[G[3],g,g,g,e]+7\,e^{-\gamma\eta}\,\eta\,U_{-1}[h,h,f,f]\,U_1[G[3],g,g,g,e]+\\
&U_{-1}[h,h,h,f,f]\,U_1[G[3],g,g,g,e]-e^{-4\gamma\eta}\,U_{-1}[h,h,h,f,f]\,U_1[G[3],g,g,g,e]-\\
&2\,e^{-3\gamma\eta}\,U_{-1}[h,h,h,f,f]\,U_1[G[3],g,g,g,e]+2\,e^{-\gamma\eta}\,U_{-1}[h,h,h,f,f]\,U_1[G[3],g,g,g,e]-\\
&e^{-5\gamma\eta}\,\eta\,U_{-1}[H[-1],h,h,f,f]\,U_1[G[4],g,g,g,e]-\\
&4\,e^{-4\gamma\eta}\,\eta\,U_{-1}[H[-1],h,h,f,f]\,U_1[G[4],g,g,g,e]-\\
&3\,e^{-3\gamma\eta}\,\eta\,U_{-1}[H[-1],h,h,f,f]\,U_1[G[4],g,g,g,e]+\\
&e^{-6\gamma\eta}\,U_{-1}[H[-1],h,h,h,f,f]\,U_1[G[4],g,g,g,e]+\\
&e^{-5\gamma\eta}\,U_{-1}[H[-1],h,h,h,f,f]\,U_1[G[4],g,g,g,e]-\\
&e^{-4\gamma\eta}\,U_{-1}[H[-1],h,h,h,f,f]\,U_1[G[4],g,g,g,e]-\\
&e^{-3\gamma\eta}\,U_{-1}[H[-1],h,h,h,f,f]\,U_1[G[4],g,g,g,e]-\\
&e^{-5\gamma\eta}\,\eta\,U_{-1}[h,h,f,f,f]\,U_1[G[2],g,g,g,e,e]+\\
&2\,e^{-4\gamma\eta}\,\eta\,U_{-1}[h,h,f,f,f]\,U_1[G[2],g,g,g,e,e]-\\
&e^{-3\gamma\eta}\,\eta\,U_{-1}[h,h,f,f,f]\,U_1[G[2],g,g,g,e,e]-\\
&4\,e^{-2\gamma\eta}\,\eta\,U_{-1}[h,h,f,f,f]\,U_1[G[2],g,g,g,e,e]+\\
&e^{-4\gamma\eta}\,U_{-1}[h,h,h,f,f,f]\,U_1[G[2],g,g,g,e,e]-\\
&e^{-2\gamma\eta}\,U_{-1}[h,h,h,f,f,f]\,U_1[G[2],g,g,g,e,e]+\\
&e^{-9\gamma\eta}\,\eta\,U_{-1}[H[-1],h,h,f,f,f]\,U_1[G[3],g,g,g,e,e]-\\
&2\,e^{-8\gamma\eta}\,\eta\,U_{-1}[H[-1],h,h,f,f,f]\,U_1[G[3],g,g,g,e,e]+\\
&e^{-7\gamma\eta}\,\eta\,U_{-1}[H[-1],h,h,f,f,f]\,U_1[G[3],g,g,g,e,e]+\\
&4\,e^{-6\gamma\eta}\,\eta\,U_{-1}[H[-1],h,h,f,f,f]\,U_1[G[3],g,g,g,e,e]-\\
&e^{-8\gamma\eta}\,U_{-1}[H[-1],h,h,h,f,f,f]\,U_1[G[3],g,g,g,e,e]+\\
&e^{-6\gamma\eta}\,U_{-1}[H[-1],h,h,h,f,f,f]\,U_1[G[3],g,g,g,e,e]
\end{aligned}
$$

## The R-Matrix

Using Quesne's formula.

```
R_{i_,j_}[d_] := Module[{x, y}, ℴ[
   U_i[x_1 → h, x_2 → f] U_i[] U_j[] U_j[y_1 → g, y_2 → e],
   Series[Exp[ħ x_1 y_1 + Sum[((1 - ℯ^(ħ η γ))^k (ħ x_2 y_2)^k)/(k (1 - ℯ^(k ħ η γ))), {k, 1, d}]], {ħ, 0, d}]
  ]]
```

$R_{1,2}[1]$

$$U_{-2}[]\,U_{-1}[]\,U_1[]\,U_2[]+\left(U_{-2}[]\,U_{-1}[f]\,U_1[]\,U_2[e]+U_{-2}[]\,U_{-1}[h]\,U_1[]\,U_2[g]\right)\hbar+O[\hbar]^2$$

```
R_{1,2}[1] // S_{-1}
```

$$U_{-2}[]\,U_{-1}[]\,U_1[]\,U_2[] + \left(-U_{-2}[]\,U_{-1}[H[-1],f]\,U_1[]\,U_2[e] - U_{-2}[]\,U_{-1}[h]\,U_1[]\,U_2[g]\right)\hbar + O[\hbar]^2$$

```
R_{1,2}[1] R_{3,4}[1] // S_{-3}
```

$$\begin{aligned}&U_{-4}[]\,U_{-3}[]\,U_{-2}[]\,U_{-1}[]\,U_1[]\,U_2[]\,U_3[]\,U_4[] + \\ &\Big(U_{-4}[]\,U_{-3}[]\,U_{-2}[]\,U_{-1}[f]\,U_1[]\,U_2[e]\,U_3[]\,U_4[] + U_{-4}[]\,U_{-3}[]\,U_{-2}[]\,U_{-1}[h]\,U_1[]\,U_2[g]\,U_3[]\,U_4[] - \\ &\quad U_{-4}[]\,U_{-3}[H[-1],f]\,U_{-2}[]\,U_{-1}[]\,U_1[]\,U_2[]\,U_3[]\,U_4[e] - \\ &\quad U_{-4}[]\,U_{-3}[h]\,U_{-2}[]\,U_{-1}[]\,U_1[]\,U_2[]\,U_3[]\,U_4[g]\Big)\hbar + O[\hbar]^2\end{aligned}$$

```
R_{1,2}[1] R_{3,4}[1] // S_4 // m_{1,3→1} // m_{2,4→2}
```

$$\begin{aligned}&U_{-4}[]\,U_{-3}[]\,U_{-2}[]\,U_{-1}[]\,U_1[]\,U_2[] + \\ &\Big(U_{-4}[]\,U_{-3}[]\,U_{-2}[]\,U_{-1}[f]\,U_1[]\,U_2[e] - U_{-4}[]\,U_{-3}[h]\,U_{-2}[]\,U_{-1}[]\,U_1[]\,U_2[g] + \\ &\quad U_{-4}[]\,U_{-3}[]\,U_{-2}[]\,U_{-1}[h]\,U_1[]\,U_2[g] - e^{\gamma\eta}\,U_{-4}[]\,U_{-3}[f]\,U_{-2}[]\,U_{-1}[]\,U_1[]\,U_2[G[-1],e]\Big)\hbar + O[\hbar]^2\end{aligned}$$

```
S_{-1}[R_{1,2}[1]] ** R_{1,2}[1]
```

$$U_{-2}[]\,U_{-1}[]\,U_1[]\,U_2[] + \left(U_{-2}[]\,U_{-1}[f]\,U_1[]\,U_2[e] - U_{-2}[]\,U_{-1}[H[-1],f]\,U_1[]\,U_2[e]\right)\hbar + O[\hbar]^2$$

```
R_{1,2}[1] ** R_{1,3}[1] ** R_{2,3}[1]
```

$$\begin{aligned}&U_{-3}[]\,U_{-2}[]\,U_{-1}[]\,U_1[]\,U_2[]\,U_3[] + \\ &\Big(U_{-3}[]\,U_{-2}[]\,U_{-1}[f]\,U_1[]\,U_2[e]\,U_3[] + U_{-3}[]\,U_{-2}[]\,U_{-1}[h]\,U_1[]\,U_2[g]\,U_3[] + \\ &\quad U_{-3}[]\,U_{-2}[f]\,U_{-1}[]\,U_1[]\,U_2[]\,U_3[e] + U_{-3}[]\,U_{-2}[]\,U_{-1}[f]\,U_1[]\,U_2[]\,U_3[e] + \\ &\quad U_{-3}[]\,U_{-2}[h]\,U_{-1}[]\,U_1[]\,U_2[]\,U_3[g] + U_{-3}[]\,U_{-2}[]\,U_{-1}[h]\,U_1[]\,U_2[]\,U_3[g]\Big)\hbar + O[\hbar]^2\end{aligned}$$

```
R_{2,3}[1] ** R_{1,3}[1] ** R_{1,2}[1]
```

$$\begin{aligned}&U_{-3}[]\,U_{-2}[]\,U_{-1}[]\,U_1[]\,U_2[]\,U_3[] + \\ &\Big(U_{-3}[]\,U_{-2}[]\,U_{-1}[f]\,U_1[]\,U_2[e]\,U_3[] + U_{-3}[]\,U_{-2}[]\,U_{-1}[h]\,U_1[]\,U_2[g]\,U_3[] + \\ &\quad U_{-3}[]\,U_{-2}[f]\,U_{-1}[]\,U_1[]\,U_2[]\,U_3[e] + U_{-3}[]\,U_{-2}[]\,U_{-1}[f]\,U_1[]\,U_2[]\,U_3[e] + \\ &\quad U_{-3}[]\,U_{-2}[h]\,U_{-1}[]\,U_1[]\,U_2[]\,U_3[g] + U_{-3}[]\,U_{-2}[]\,U_{-1}[h]\,U_1[]\,U_2[]\,U_3[g]\Big)\hbar + O[\hbar]^2\end{aligned}$$

```
With[{d = 1}, R_{1,2}[d] ** R_{1,3}[d] ** R_{2,3}[d] == R_{2,3}[d] ** R_{1,3}[d] ** R_{1,2}[d]]
```

```
True
```

```
With[{d = 2}, USimp[R_{1,2}[d] ** R_{1,3}[d] ** R_{2,3}[d] - R_{2,3}[d] ** R_{1,3}[d] ** R_{1,2}[d]]]
```

$$\begin{aligned}&\Big(-\gamma\,U_{-3}[]\,U_{-2}[h]\,U_{-1}[f]\,U_1[]\,U_2[]\,U_3[e] + \\ &\quad U_{-3}[]\,U_{-2}[H[1]]\,U_{-1}[f]\,U_1[]\,U_2[]\,U_3[e] - \eta\,U_{-3}[]\,U_{-2}[]\,U_{-1}[f]\,U_1[]\,U_2[g]\,U_3[e] - \\ &\quad U_{-3}[]\,U_{-2}[]\,U_{-1}[f]\,U_1[]\,U_2[G[1]]\,U_3[e] + 2\,\eta\,U_{-3}[]\,U_{-2}[]\,U_{-1}[f]\,U_1[]\,U_2[e]\,U_3[g]\Big)\hbar^2 + O[\hbar]^3\end{aligned}$$